# $Id: PKGBUILD 135992 2011-08-21 06:01:59Z eric $
# Maintainer: Aaron Griffin <aaron@archlinux.org>
# Contributor: Paul Mattal <pjmattal@elys.com>

pkgname=fbset
pkgver=2.1
pkgrel=4
pkgdesc="Framebuffer setup utility"
arch=('i686' 'x86_64')
url="http://users.telenet.be/geertu/Linux/fbdev/"
license=('GPL')
depends=('glibc')
backup=('etc/fb.modes')
options=('!makeflags')
source=(http://users.telenet.be/geertu/Linux/fbdev/fbset-${pkgver}.tar.gz)
md5sums=('e547cfcbb8c1a4f2a6b8ba4acb8b7164')

build() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  make
}

package() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  install -D -m755 fbset "${pkgdir}/usr/sbin/fbset"
  install -D -m644 fb.modes.5 "${pkgdir}/usr/share/man/man5/fb.modes.5"
  install -D -m644 fbset.8 "${pkgdir}/usr/share/man/man8/fbset.8"
  install -D -m644 etc/fb.modes.ATI "${pkgdir}/etc/fb.modes"
}
